
Thème 1 - TP 1 : fonction de  dans , expression algébrique, représentation graphique, image réciproque d’un élément par la fonction.

Application à la densité de la loi normale et ses quantiles

 

Sujet

1. Corrigé, approche élémentaire

2. Résolution numérique d’une équation

3. Fonction de répartition

4. Programmation avancée

Complément variables, expressions, fonctions, affectations

Sujet

On se souvient que la densité de la loi normale centrée est de la forme  où a et b sont tels que l’intégrale sur  vaut 1. Trouver la relation entre a et b puis exprimer a et b en fonction de
l’écart type ; donner l’expression de f avec comme paramètre l’écart type. Comment obtient on la densité associée à une espérance m au lieu de 0 ?

Dessiner le graphe de f pour différentes valeurs de l’espérance et de l’écart type.

Trouver a tel que l’intégrale de f sur [-a,a] vaut 0.95 (Cf intervalle de confiance à 95%).

Au passage : recherche de documentation, éléments de programmation en Python : déclaration - définition de variables, affectation, expressions algébriques, valeur numérique, substitution dans une
expression, fonctions (voir cette page (en))

Un corrigé

Distinguer les difficultés de syntaxe (syntaxe Python), de programmation (méthodes), de documentations, algorithmiques, mathématiques : faire la part des choses

1. Approche élémentaire (le moins d’astuce de programmation que possible)

2 #faire une recherche sur "sagemath intégration", menu Calculus de la feuille interactive SageMathCloud, ...
3 integrate(exp(-x^2),x,0,1)
4 integrate(exp(-x^2),x)
5 I=integrate(exp(-x^2),x,-infinity,infinity);I

1/2*sqrt(pi)*erf(1) 
1/2*sqrt(pi)*erf(x) 
sqrt(pi) 

6 I.n?
7 #commande suivi de ? pour une documentation sur la commande. 1ères lettres de la commande suivi de <tab> pour listes des commandes de préfixe donné ; de même objet suivi de . et <tab> pour liste des méthodes s'appliquant à l'objet
8 #recherche sur les mots clef "sagemath valeur numérique"
9 #numerical_approx? erf?

File: /projects/sage/sage-7.5/src/sage/structure/element.pyx 
Signature : I.n(self, prec=None, digits=None, algorithm=None) 
Docstring : 
Alias for "numerical_approx()". 

EXAMPLES: 

   sage: (2/3).n() 
   0.666666666666667

10 integrate(exp(-x^2),x,0,1).n(digits=3)
11 I.n(digits=3)
12 numerical_approx(I,digits=3)

0.747 
1.77 
1.77 

13 reset() # supprime les variables déjà déclarées ou affectées ; la réexécution produira le même résultat
14 #L'intégrale comme fonction de paramètres :
15 var('a b');III=integrate(a*exp(-(x*b)^2),x,-infinity,infinity,hold=true);III
16 #sans déclaration des paramètres autres que x, la commande donnerait une erreur. Cf var? , show_identifiers()
17 #hold=true pour que III renvoie l'expression littérale et non celle obtenue après calcul
18 #sans hold=true, le calcul produit une erreur, invite à précéder le calcul de assume(b>0)
19 #
20 III.parent() #donne la structure dont III est un élément, III n'est pas une fonction

R R

R R

f(x) = a exp(−b )x2
R

https://cloud.sagemath.com/projects/a56d8802-5740-429a-b5da-e7eb8dc5bb82/files/1.sagews
data:application/octet-stream;base64,
http://doc.sagemath.org/html/en/thematic_tutorials/index.html

21 #
22 #Comparer avec :
23 reset();print
24 II(a,b)=integrate(a*exp(-(x*b)^2),x,-infinity,infinity,hold=true)
25 #cette commande déclare II comme fonction et ses arguments a,b
26 show_identifiers()
27 II
28 II.parent()
29 II(1,1)#hold=true est oublié, comparer avec integrate(exp(-(x)^2),x,-infinity,infinity,hold=true)

(a, b) 
integrate(a*e^(-b^2*x^2), x, -Infinity, +Infinity) 
Symbolic Ring 
 
['a', 'II', 'b'] 
(a, b) |--> integrate(a*e^(-b^2*x^2), x, -Infinity, +Infinity) 
Callable function ring with arguments (a, b) 
sqrt(pi) 

30 show(II) #Visualisation LaTeX de l'expression II

31 assume(b>0);simplify(II)
32 assumptions()
33 #sans assume(b>0) la commande simplify(II) produit une erreur, le message d'erreur invite à utiliser simplify
34 #assume(b>0) reste actif pour les cellules suivantes. Pour revenir en arrière forget(b>0), pour tout oublier forget()
35 #cf assume? assuptions? forget?

(a, b) |--> sqrt(pi)*a/b 
[b > 0] 

36 #relation entre a et b pour que II(a,b)=1, cf recherche sur mot clef sagemath solution équation
37 #http://doc.sagemath.org/html/fr/tutorial/tour_algebra.html
38 solve(II(a,b)==1,a) #assume(b>0) implicite, sans quoi une erreur est produite

[a == b/sqrt(pi)] 

39 #densité gaussienne centrée
40 a = b/sqrt(pi);f(x)=a*e^(-b^2*x^2)
41 f
42 integrate(f(x),x,-infinity,infinity)

x |--> b*e^(-b^2*x^2)/sqrt(pi) 
1 

43 #Espérance
44 E1=integrate(x*f(x),x,-infinity, infinity);E1

0 

45 #Variance
46 E2=integrate(x^2*f(x),x,-infinity, infinity);V=E2-E1^2;V
47 show(V)

1/2/b^2 

48 #b en fonction de l'écart-type puis expression de f
49 var('s');solve(V==s^2,b)

s 
[b == -1/2*sqrt(2)/s, b == 1/2*sqrt(2)/s] 

50 #substitution à la main dans l'expression de f
51 b = 1/2*sqrt(2)/s
52 f(x)=b*e^(-b^2*x^2)/sqrt(pi)
53 f
54 show(f)

x |--> 1/2*sqrt(2)*e^(-1/2*x^2/s^2)/(sqrt(pi)*s) 

55 #Représentation graphique
56 #s=1;plot(f) produit une erreur : s=1 n'est pas pris en compte dans l'évaluation de f. La bonne méthode est f(s=1) cf "sagemath substitution expression"
57 s=1;f(1)
58 f(s=1)
59 g(x)=f(s=1)
60 g
61 plot(g)
62 #google "sagemath plot" ou plot?
63 plot(g,-5,5)

1/2*sqrt(2)*e^(-1/2/s^2)/(sqrt(pi)*s) 
1/2*sqrt(2)*e^(-1/2*x^2)/sqrt(pi) 
x |--> 1/2*sqrt(2)*e^(-1/2*x^2)/sqrt(pi) 

(a, b)  ↦  a dx∫ +∞

−∞

e(− )b2x2

1

2 b2

x ↦ 
2–√ e

(− )x2

2 s2

2 sπ−−√

64 #densité gaussienne centrée en 1, d'écart type 1/2
65 g(x)=f(s=1/2,x=x-1) #drole de syntaxe !
66 g
67 plot(g,-5,5)

x |--> sqrt(2)*e^(-2*(x - 1)^2)/sqrt(pi) 

68 #a en fonction de s tel que l'intégrale de f sur [-a,a] vaut 0.95
69 var('a');integrate(f(x),x,-a,a)

a 
erf(1/2*sqrt(2)*a/s) 

70 var('s') #réinitialisation de s
71 solve(erf(1/2*sqrt(2)*a/s)==0.95,a)

s 
[a == sqrt(2)*s*inverse_erf(19/20)] 

72 #a est proportionnel à s, ça se voyait déjà sur l'équation
73 #valeur numérique de a pour s=1 ?
74 s=1;(sqrt(2)*s*inverse_erf(19/20)).n()

Error in lines 1-1 
Traceback (most recent call last): 
  File "/projects/sage/sage-7.5/local/lib/python2.7/site-packages/smc_sagews/sage_server.py", line 995, in execute 
    exec compile(block+'\n', '', 'single') in namespace, locals 
  File "", line 1, in  
NameError: name 'inverse_erf' is not defined 

*** WARNING: Code contains non-ascii characters *** 

75 #inverse_erf donné dans une réponse par Sage n'est pas (en mai 2017) un objet de Sage
76 #voir à ce sujet https://ask.sagemath.org/question/8118/evaluating-inverse-erf/
77 #Une version numérique de inverse_erf peut être importée de la librairie Python Scipy, cf recherche sur mots clef "python inverse erf"
78 from scipy.special import erfinv as inverse_erf
79 inverse_erf?
80 #On revient en arrière par la commande
81 #del inverse_erf

File: /projects/sage/sage-7.5/local/lib/python2.7/site-packages/scipy/special/basic.py 
Signature : inverse_erf(y) 
Docstring : 
Inverse function for erf.

82 s=1;(sqrt(2)*s*inverse_erf(19/20)).n()

1.95996398454005 

2. Résolution numérique d’une équation

83 #Méthode implémentée dans Python, cf find_root?
84 #Recherche d'une valeur approchée de a entre 0 et 4
85 find_root(erf(1/2*sqrt(2)*a)==0.95,0,4)

1.959963984540053 

86 #Algorithme de dichotomie : recherche d'un 0 de la fonction h entre a et b à la précision prec si h change de signe entre a et b 
87 def dicho(h,a,b,prec):
88     if h(a)*h(b)>0:
89         print('peut être pas de solution')
90     else:
91         while b-a>prec:
92             i=(a+b)/2
93             if h(i)*h(b)>0:b=i
94             else:a=i
95     return(a)

96 h(x)=erf(1/2*sqrt(2)*x)-0.95
97 dicho(h,0,4,0.001).n() #seules les 3 premières décimales après la virgule sont pertinentes !

1.95996093750000 

98 #Vérification
99 show(integrate(f(s=1),x,-1.96.n(digits=3),1.96.n(digits=3),hold=true),'=',integrate(f(s=1),x,-1.96,1.96).n())

 = 

3. Fonction de répartition

100 #Fonction de répartition de la densité gaussienne centrée réduite (s=1)
101 F(t)=integrate(f(s=1),x,-infinity,t);F

t |--> 1/4*sqrt(2)*(sqrt(2)*sqrt(pi)*erf(1/2*sqrt(2)*t) + sqrt(2)*sqrt(pi))/sqrt(pi) 

102 #Méthode simplify disponible pour F (faire F. suivi de <tab>)
103 F.simplify_full() #bizarement on obtient F(t) plutôt que F
104 F(t)=F.simplify_full();F

1/2*erf(1/2*sqrt(2)*t) + 1/2 
t |--> 1/2*erf(1/2*sqrt(2)*t) + 1/2 

105 plot(F,-5,5)

106 #On cherche a tel que F(a)-F(-a)=0.95 soit F(a)=0.975, d'abord sur le dessin
107 #1.96 est choisi "à la main" et on vérifie par le tracé des lignes en pointillés que c'est une bonne approximation
108 #ticks en option : voir http://doc.sagemath.org/html/en/reference/plotting/sage/plot/plot.html
109 plot(F,1.8,2.2)+line([(1.8,0.975),(2,0.975)],linestyle='--',ticks=[[1.8,1.9,1.96,2,2.1,2.2],None])+line([(1.96,0.95),(1.96,F(2.2))],linestyle='--')

dx∫ 1.96

−1.96

2–√ e
(− )1

2
x2

2 π−−√
0.950004209703559

110 #Méthode numérique avec find_root
111 find_root(F(x)==0.975,0,4)

1.959963984540053 

112 #Méthode formelle
113 solve(F(x)==0.975,x) #même solution que précédemment, même pb avec inverse_erf

[x == sqrt(2)*inverse_erf(19/20)] 

4. Programmation plus efficace. Qu’est qui change par rapport à la section 1. ?

114 show_identifiers()
115 reset();print;show_identifiers()

['g', 'dicho', 'V', 'F', 'h', 's', 't', 'inverse_erf', 'a', 'E1', 'E2', 'II', 'b', 'f'] 
 
[] 

116 var('a','b')
117 f(x)=a*exp(-(b*x)^2);f
118 var('m')
119 assume(b != 0);I=integrate(f(x-m),x,-infinity,infinity);I #sans assume(b != 0) une erreur est produite dont le texte invite à une telle commande

(a, b) 
x |--> a*e^(-b^2*x^2) 
m 
sqrt(pi)*a/sqrt(b^2) 

120 assume(b>0);simplify(I)

sqrt(pi)*a/b 

121 S=solve(I==1,a);S

[a == b/sqrt(pi)] 

122 f=f.subs(S[0]);f
123 #alternative : f=f(a=S[0].rhs())

x |--> b*e^(-b^2*x^2)/sqrt(pi) 

124 E1=integrate(x*f(x-m),x,-infinity,infinity);E1 #espérance
125 E2=integrate(x^2*f(x-m),x,-infinity,infinity);V=E2-E1^2;V #variance
126 V=V.simplify_full();V #V ne devrait pas dépendre de m !
127 show('V = ',V)

m 
-m^2 + 1/2*(2*sqrt(pi)*m^2/b + sqrt(pi)/b^3)*b/sqrt(pi) 
1/2/b^2 

V = 

128 var('s');B=solve(s^2==V,b);B

s 
[b == -1/2*sqrt(2)/s, b == 1/2*sqrt(2)/s] 

129 f=f.subs(B[1]);show('f_s (x) = ',f)

f_s (x) = 

130 la='$'+latex(f)+'$'
131 html('<h3><div align="center"> $f_s(x)\ =\ $'+la+'</div></h3>')

 

1

2 b2

x ↦ 
2
–

√ e
(− )x2

2 s2

2 sπ−−√

(x)  =  fs x ↦ 
2√ e

(− )x2

2 s2

2 sπ√

132 #densité gaussienne centrée en m
133 g(m,s,x)=f(x-m)
134 show(g)

135 #Représentation graphique, cf "sagemath graphique", plot?
136 m=1;s=1
137 plot(g(m,s,x),x,-4,6)+line([(m,0),(m,g(m,s,m))],linestyle='--')+text("m",(m,-0.015))

138 c = ['blue','green','red','cyan','magenta','yellow','black']
139 m=[-1,0,1,1];s=[1,1,1/2,2]
140 p=sum(plot(g(m[k-1],s[k-1],x),x,-5,5, color=c[k-1],legend_label='(m,s)='+str((m[k-1],s[k-1]))) for k in (1..4))
141 show(p)

142 #Recherche d'un quantile pour s=1
143 #Méthode formelle avec la fonction de répartition
144 from scipy.special import erfinv as inverse_erf
145 F(t)=integrate(g(0,1,x),x,-infinity,t)
146 solq=solve(F(x)==1.95/2,x);solq
147 x.subs(solq) #la définition scipy de inverse_erf n'affecte pas la méthode solve
148 #x.subs(solq).n() produit une erreur ; on recopie à la main le texte de la solution ; peut on le faire par une commande à partir de str(solq[0]) ? cf eval (python -> 1/2=0), sage_eval (-> ne reconnait pas inverse_erf)
149 x.subs(x == sqrt(2)*inverse_erf(19/20)).n(digits=3)

[x == sqrt(2)*inverse_erf(19/20)] 
sqrt(2)*inverse_erf(19/20) 
1.96 

150 #test
151 %python
152 import platform
153 platform.python_version()
154 1/2 #le résultat est un entier dans python 2.* ! cf "python division"
155 1./2
156 2^3 #Cf la section 9.9.1 de https://docs.python.org/2/library/operator.html
157 2**3

'2.7.13' 
0 
0.5 
1 
8 

158 %python3
159 import platform
160 print(platform.python_version())
161 print(1/2)

(m, s,x)  ↦ 
2–√ e

(− )(m−x)2

2 s2

2 sπ−−√

3.5.3 
0.5 

162 #test dans sage
163 from scipy.special import erfinv as inverse_erf
164 eval('1/2')
165 eval('1./2')
166 sage_eval('1/2')
167 #sage_eval('inverse_erf(1/2)') produit une erreur : inverse_erf pas reconnu
168 eval('inverse_erf(1/2)')
169 eval('inverse_erf(1./2)')

0 
0.5 
1/2 
0.0 
0.47693627620446982 

170 eval(str(solq[0].rhs()).replace('/','*1.0/')).n()

1.95996398454005 

171 #Algorithme de dichotomie
172 def dicho(h,a,b,prec):
173     if h(a)*h(b)>0:
174         print('peut être pas de solution')
175     else:
176         while b-a>prec:
177             i=(a+b)/2
178             if h(i)*h(b)>0:b=i
179             else:a=i
180     return(a)

181 dicho(F-0.975,0,4,0.001).n() #seules les 3 premières décimales après la virgule sont pertinentes !

1.95996093750000 

Compléments : variables, expressions, fonctions, affectations dans Python

182 #Test
183 reset()
184 f=x;f #x est prédéclaré ; f=y produit une erreur si y n'est pas déclaré au préalable
185 x=1;f #pas rétroactif
186 f=x;f #x est remplacé par sa valeur
187 f(x)=x;f;x#x est réinitialisé !
188 print
189 f=x;f
190 f.parent()
191 f.variables()
192 f(x=1) #substitution de x par 1 dans l'expression f
193 show_identifiers() #x est omis par convention
194 print
195 g=f;g
196 x=1
197 diff(f,x)
198 x=var('x');diff(f)#avec x=1 produit une erreur
199 integrate(f)
200 var('y');integrate(f(x=y),y)#avec y=1 produit une erreur
201 print
202 g(x)=f;print g,',', g.parent(),',',g.variables()
203 #g(y)=f.subs(x=y);g
204 g.diff()
205 diff(g(y),y)
206 integrate(g)
207 integrate(g(y),y)

x 
x 
1 
x |--> x 
x 
 
x 
Symbolic Ring 
(x,) 
1 
['f'] 
 
x 
1 
1 
1/2*x^2 
y 
1/2*y^2 
 
x |--> x , Callable function ring with argument x , (x,) 
x |--> 1 
1 
x |--> 1/2*x^2 
1/2*y^2 

208 reset()
209 h=lambda x:x^2;h;h(2);h(x);diff(h(x),x);integrate(h(x),x)
210 #mais diff(h) produit une erreur, de même que h.variables()

211 print
212 def k(x):return x^3
213 k;k(x);diff(k(x),x)
214 show_identifiers()
215 plot(h)+plot(k)

 at 0x7f10bb28db18> 
4 
x^2 
2*x 
1/3*x^3 
 
 
x^3 
3*x^2 
['k', 'h'] 

216 reset()
217 #y n'est pas défini, une expression contenant y produit une erreur, mais pas une affectation
218 g(y)=x;g;g(1) #déclare g et y
219 show_identifiers()
220 g.parent()

y |--> x 
x 
['g', 'y'] 
Callable function ring with argument y 

221 #variables locales dans def, affectation globale
222 reset()
223 x=2
224 def g(y):
225     z=y
226     y=1
227     x=1
228     return(z)
229 show_identifiers()
230 g(x)
231 x

['g', 'x'] 
2 
2 

232 str(g(y))

'x' 

233 import ast
234 formula = 'integrate(y+a^2,a,0,1)'
235 names = [
236     node.id for node in ast.walk(ast.parse(formula)) 
237     if isinstance(node, ast.Name)
238 ]
239 names

['integrate', 'a', 'y', 'a'] 

240 del ast

